
Demystifying Parametric Analyses: Illustrating
Canonical Correlation Analysis as the

Multivariate General Linear Model

Robin K. Henson, University of Southern Mississippi
A review of the research literature suggests that teachers need to provide students with engaging problems, facilitate
their discovery of analysis methods, and encourage classroom discussion and presentation of their approaches to
solving problems.  The present article illustrates how canonical correlation analysis can be employed to implement
all the parametric tests that canonical methods subsume as special cases, including multiple regression.  The point is
heuristic: all analyses are correlational, all apply weights to measured variables to create synthetic variables, and all
yield effect sizes analogous to r2.  Knowledge of such relationships helps inform researcher judgement of analysis
selection and use.  

n one of his seminal contributions, the late Jacob
“Jack” Cohen (1968) demonstrated that multiple
regression subsumes all the univariate parametric

methods as special cases, and thus provides a
univariate general linear model (GLM) that can be
employed in all univariate analyses.  At about the
same time, researchers increasingly also came to
realize that ANOVA was being overused, and in many
cases used when other methods would have been more
useful.  One source of ANOVA overuse was that too
many researchers erroneously associated ANOVA as an
analysis with the ability to make causal statements
when using experimental research designs; however, it
is the design, and not the analysis that leads to the
ability to make definitive causal statements!

As Humphreys (1978) explained this
phenomonon:

The basic fact is that a measure of
individual differences is not an independent
variable [in an experimental design], and it
does not become one by categorizing the
scores and treating the categories as if they
defined a variable under experimental
control in a factorially designed analysis of
variance. (p. 873, emphasis added)

Similarly, Humphreys and Fleishman (1974) noted
that categorizing variables in a nonexperimental design
using an ANOVA analysis “not infrequently produces
in both the investigator and his audience the illusion
that he has experimental control over the independent
variable.  Nothing could be more wrong” (p. 468).

Furthermore, as Cliff (1987) noted, the practice of
discarding variance on intervally-scaled predictor
variables in order to perform ANOVA-type analyses
creates problems in almost all cases:

Such divisions are not infallible; think of
the persons near the borders.  Some who
should be highs are actually classified as
lows, and vice versa.  In addition, the
“barely highs” are classified the same as the
“very highs,” even though they are
different.  Therefore, reducing a reliable

variable to a dichotomy makes the variable
more unreliable, not less. (p. 130,
emphasis added)

These various realizations have led to less frequent use
of ANOVA methods, and to more frequent use of
general linear model approaches such as regression (cf.
Edgington, 1974; Elmore & Woehlke, 1988; Goodwin
& Goodwin, 1985; Willson, 1980).

Since all analyses are correlational, and it is the
design and not the analysis that yields the capacity to
make causal inferences, the practice of converting
intervally-scaled predictor variables to nominal scale so
that ANOVA and other OVAs (i.e., ANCOVA,
MANOVA, MANCOVA) can be conducted is
inexcusable in many cases.

However, canonical correlation analysis, and not
regression analysis, is the most general case of the
general linear model (Baggaley, 1981; Fornell, 1978;
Thompson, 1991, 1998).  [Structural equation
modeling (SEM) represents an even broader general
linear model, but SEM is somewhat different in that
this analysis usually also incorporates measurement
error estimation as part of the analysis (cf. Bagozzi,
Fornell, & Larcker, 1981; Fan, 1996, 1997).]  In an
important article, Knapp (1978) demonstrated this in
some detail and concluded that “virtually all of the
commonly encountered parametric tests of significance
can be treated as special cases of canonical correlation
analysis” (p. 410).

The present article will illustrate how canonical
correlation analysis can be employed to implement all
the parametric tests that canonical methods subsume as
special cases.  The point is not that all research ought
to be conducted with canonical analyses, rather the
point is heuristic: all analyses are correlational, all
analyses apply weights to measured variables to create
synthetic variables that become the analytic focus, and
all yield effect sizes analogous to r2 that are important
to interpret.  For example the R2 obtained in a
multiple regression, the eta2 obtained from an
ANOVA, and the squared canonical correlation
coefficient obtained from a canonical correlation

analysis all describe the variance-accounted-for between
two variables and/or sets of variables.  Ultimately,
these statistics are directly analogous to the squared
Pearson correlation.

Understanding general linear model principles aids
in realizing that parametric analyses are all
fundamentally related.  Individual methods, such as
ANOVA or t-tests, can then be viewed from a global
perspective which will, hopefully, facilitate thoughtful
researcher judgment in selecting analyses as opposed to
employing “lock-step” decision strategies that limit
the utility of analyses.

The Basics of Canonical
Correlation Analysis

While a comprehensive discussion of CCA is
beyond a scope of the present article, the reader is
referred to Thompson (1991) for an accessible and user-
friendly treatment of CCA.  Furthermore, neither the
analytic derivations of CCA nor the equivalent
derivations of the linear models for the various
analyses will be visited here.  Since the purpose of
this article is to demonstrate equivalence of models
through obtained results, the reader is referred to Knapp
(1978) for mathematical demonstration of the linear
models.

The theory of canonical correlation analysis
(CCA) has been with us for considerable time
(Hotelling, 1935), but did not come into practical use
until the onset of computerization (Krus, Reynolds, &
Krus, 1976).  In canonical analysis, the variables are
considered to be members of two or more (in practice,
almost always two) variable sets (e.g., pretest and
posttest scores, aptitude and achievement scores) –
otherwise we would analyze the data with factor
analysis so as to consider simultaneously all the
relationships, but without considering the existence of
variable sets.  Each set will include more than one
variable, otherwise we generally would use a Pearson   r  
or regression analysis.  As will be shown later, these
analyses are essentially the same thing anyway!

A CCA will yield many useful statistics, the
most recognized of which is the canonical correlation
(Rc).  The canonical correlation describes the
relationship between two synthetic variables that have
been modeled from their respective variable sets by
applying weights to the measured variables.  A
canonical correlation will be produced for each function
(i.e., for each set of standardized canonical function
coefficients and respective measured variables).  The
number of functions, each of which will be perfectly
uncorrelated with the others, equals the number of
variables in the smaller of the variable sets.  The
canonical correlation can be squared to yield a variance-
accounted-for effect size (Rc

2), or the percentage of
variance explainable in the criterion variable set
predictable with knowledge of the variance in the
predictor set.

One advantage of CCA, and other multivariate
methods, lies in its simultaneous examination of the
variables of interest, thus reducing risk of
experimentwise Type I error (Fish, 1988; Henson, in
press; Thompson, in press).  A second, and perhaps
often overlooked, advantage is the flexibility of the
analysis in looking at various research problems.  One
example of this versatility can be found in a
measurement study involving multivariate criterion-
related score validity (Sexton, McLean, Boyd,
Thompson, & McCormick, 1988).  Thus, CCA can
be used in either substantive or measurement inquiries.

Canonical Correlation Analysis as
the General Linear Model

An heuristic data set for 12 elementary, middle,
and high school students will be used to illustrate that
CCA can conduct the other parametric methods that it
subsumes, both univariate and multivariate alike.
CCA will be used to perform a t-test, Pearson
correlation, multiple regression, ANOVA, MANOVA,
and descriptive discriminant analysis.  Table 1 lists
heuristic data on four intervally scaled variables related
to motivational and personality issues: attributions of
effort (EFFORT), attributions of ability (ABILIT),
locus of control (LOCUS), and degree of extroversion
(EXTROV). Also included are grouping data indicating
some experimental treatment (TREAT) and whether
students are in elementary, middle, or high school
(GRADE).  The reader will also notice five planned
contrast variables which will be described later.

Analyses will be run using the SPSS (v9.0)
statistics package.  The canonical correlation macro
(CANCORR) is a new addition to this version of
SPSS but it limits analyses to two sets of variables of
equal size.  Since several examples used here include
analyses on variable sets of differing size, the
canonical macro was not used.  There is also a General
Linear Model menu which can be used to run a variety
of analyses.  However, for the sake of consistency and
clarity, a uniform command syntax will be used in the
present article to illustrate the relationship between
canonical correlation the other parametric analyses.
This command syntax is included in the Appendix.
Note that CCA is conducted here using the MANOVA
command (again, suggesting that these analyses are be
related).  Using Table 1 variable names, the SPSS
commands for CCA are:

    MANOVA
LOCUS EXTROV WITH EFFORT ABILIT
/PRINT=SIGNIF (MULTIV EIGEN DIMENR)
/DISCRIM=(STAN ESTIM COR ALPHA(.99)).

The SAS statistical software has a more direct
command for CCA: PROC CANCORR.  An example
of SAS syntax used to perform a similar heuristic
illustration can be found in Campbell and Taylor
(1996).      

Table 1. Heuristic Data (n=12) for Canonical Correlation Illustration

ID EFFORT ABILIT LOCUS EXTROV GRADE TREAT CGR1 CGR2 CTREAT CTGR1 CTGR2
1 10 12 18 15 1 1 -1 -1  -1  1  1
2 15 14 19 16 1 1 -1 -1  -1  1  1
3 17 18 18 13 1 2 -1 -1   1 -1 -1
4 14 13 15 10 1 2 -1 -1   1 -1 -1
5 09 15 14 04 2 1  0  2  -1  0 -2
6 06 19 16 04 2 1  0  2  -1  0 -2
7 06 20 12 07 2 2  0  2   1  0  2
8 07 19 16 03 2 2  0  2   1  0  2
9 18 11 06 18 3 1  1 -1  -1 -1  1

10 17 10 04 13 3 1  1 -1  -1 -1  1
11 12 09 10 12 3 2  1 -1   1  1 -1
12 14 13 09 14 3 2  1 -1   1  1 -1

Conducting Pearson Correlation
with Canonical Correlation

When examining relationships between two
variables, a Pearson correlation (r) is often invoked.
The reader should immediately note conceptual
similarities between a Pearson r and canonical
analysis, even before examining the results from the
SPSS analysis.  Both investigate relationships
between variables, only in the canonical case the
measured variables of interest occur within
multivariate sets.

A Pearson r was computed for EFFORT and
ABILIT.  Table 2 reports the obtained results, r = -
.6150, p = .033.  Table 2 also reports the CCA
results, including the canonical correlation (Rc),
squared canonical correlation (Rc

2), and Wilks lambda
(λ) .  Wilks lambda, like Rc

2 is a variance-accounted-
for type statistic.  However, Wilks lambda indicates
the variance not accounted for in the canonical
correlation, modeled by (1 – Rc

2).  It is used for testing
the statistical significance of Rc.  As the magnitude of
_ decreases (ranging from 0 to 1), the effect size (Rc

2)
increases as does the likelihood of obtaining statistical
significance.

For these variables, the CCA computed a squared
canonical correlation coefficient of .378.  The simple
square root transformation of Rc

2 = .378 gives us R c =
.6148.  The Pearson r and canonical correlation values
are identical, save for rounding error and the fact that a
canonical correlation cannot be negative.  This is
because the weights that are used in CCA scale the
variables in the same direction, as such R c will always
range from 0 to 1. The p values are identical.

Herein lies the most fundamental of general linear
model principles: all analyses are correlational.  The
canonical correlation is nothing more than a bivariate r
between the synthetic variables created in CCA after
the application of weights.  As Thompson (1991)
noted, “This conceptualization is appealing, because
most researchers feel very comfortable thinking in
terms of the familiar bivariate correlation coefficient”
(p. 81).

Since the present heuristic CCA only had one
variable in each set, the synthetic variables reflected
the same relationship as did a Pearson r between the
variables without the application of weights.  This
result should not be surprising, given the fact that
multiplicative constants do not affect the value of r.
The only effect the weights had in this case was to
scale the variables in the same direction, thus yielding
a positive value for Rc.

Conducting Multiple Regression with
Canonical Correlation

As Cohen (1968) indicated, multiple regression
subsumes all other univariate parametric analyses as
special cases.  Therefore, there is a directly analogous
relationship between Pearson r and multiple
regression.  Since CCA subsumes Pearson r, it should
be apparent that it will do the same for multiple
regression.

A multiple regression analysis was conducted with
EFFORT predicted by LOCUS and EXTROV.  SPSS
results of the regression and canonical analyses are
found in Table 3.  Again, all parallel statistics match
within rounding error, with the exception of the
weights.  However, the difference between the weights
is arbitrary at this point.  Beta (B) weights and
standardized function coefficients are easily converted
into each other using the following formulas
(Thompson, in press):

B / Rc = Function Coefficient
Function Coefficient * R = B
For example, LOCUS had a B weight of -

.171156.  Using R c = .828 from the CCA, we find
that the standardized function coefficient matches,
within rounding error, that reported in Table 3 (-
.171156 / .828 = -.2067).  Since we know from the
obtained results that the regression multiple R  equals
the canonical R c, we can use the conversion formulas
to find canonical function coefficients using only a
regression analysis and B weights using only CCA.

Table 2. Conducting Pearson Correlation with
Canonical (EFFORT by ABILITY)

Pearson r Analysis Canonical Analysis
r -.615 R c .615
r2 .378 Rc

2 .378
lambda .622

p .033 p .033
Note. Rc cannot be negative.

Also of note here is the relationship between a
Pearson r, the obtained multiple R  from the
regression, and the Rc from the canonical analysis.  A
regression analysis applies weights to observed
(manifest) predictor variables to create a synthetic
variable called predicted Y (or sometimes YHAT), which
is a linear combination of the predictor variables.  The
multiple R  from the regression analysis is nothing
more than a Pearson correlation between predicted Y
and the observed dependent measure, EFFORT in this
case (R predicted Y, EFFORT).  Furthermore, as shown
above, the canonical correlation (Rc) also is a Pearson
r between two synthetic variables.  In this case,
however, only the predictor set (LOCUS and
EXTROV) was linearly combined via the application
of weights.  While technically the dependent measure
(EFFORT) also was transformed by a multiplicative
weight, since only one variable existed, the weight
was +1 and the EFFORT variable did not change.  As
such, the CCA and the multiple regression yielded
identical results, both of which are based on a simple
Pearson r between two variables (either manifest or
synthetic)!  

Conducting t-test and Point-biserial
Correlation with Canonical Correlation
One of the most basic of statistical analyses is the

t-test which is used to compare means between groups.
Here a t-test was used to evaluate if the treatment and
control groups (TREAT) differed on the EFFORT
variable.  Results reported in Table 4 indicate that the
means of the groups were not statistically significantly
different, t = .310, p = .760.  A canonical analysis on
the same variables yielded F(1, 10) = .100, p = .760.
Note that the p calculated values are identical between
analyses.  The test statistics (t and F) are different only
in metric.  In fact, the F distribution consists of
squared values of the t distribution.  Squaring t = .310
produces .096 which does match the F value.  The
slight difference in the values is arbitrary and solely
due to rounding error by the statistics program.

A point-biserial correlation was also conducted to
illustrate the correlational nature of even the t-test.  In
essence, a t-test is can be conceptualized as a
correlation between one dichotomous variable
(TREAT) which indicates group membership and one
continuous variable (EFFORT) as the dependent

Table 3. Conducting Multiple Regression with
Canonical (EFFORT by LOCUS and ABILIT)

Regression Analysis Canonical Analysis
R .828 R c .828
R2 .685 Rc

2 .685
lambda .315

F(2, 9) 9.797 F(2, 9) 9.797
p .006 p .006

Beta Weights Function Coefficients
LOCUS -.171 LOCUS -.207

EXTROV .767 EXTROV .926

measure.  The point-biserial correlation is a
generalization of the Pearson r illustrated above that
allows for a dichotomy in one of the variables.  Again
looking at Table 4, we see that the p values are
identical across the t-test, point-biserial, and canonical
analyses.  Furthermore, the point-biserial correlation
matches the magnitude of the canonical correlation
within rounding error.  Remember that a canonical
correlation cannot be negative as discussed above.  The
point is again made here that all analyses are
correlation in nature, even those which utilize
dichotomous variables.

Conducting Factorial ANOVA with
Canonical Correlation

The SPSS syntax file (see Appendix) includes
commands to compute the five orthogonal contrast
variables reported in the Table 1 data.  Planned
contrasts can be used with ANOVA methods to test
specific, theory-driven hypotheses as against omnibus
hypotheses (Thompson, 1994).  One advantage of
using planned contrasts is the ease of pinpointing
statistically significant effects without having to
conduct post-hoc tests which include Bonferroni-type
corrections for experimentwise error.  It is important
to note that the contrasts will yield the same overall
effect [i.e., Sum of Squares (SS) explained] as the
omnibus test.  They are necessary here to show that
CCA can conduct ANOVA.

In the present analysis, a 3 X 2 factorial ANOVA
was conducted with TREAT and GRADE as
independent variables and EFFORT as the dependent
variable.  For the CCA, the contrast variables from
Table 1 were used.  The total number of orthogonal
contrasts that can be created equals the degrees of
freedom for each main effect.  The GRADE main effect
has two degrees of freedom and is represented by
CGR1 and CGR2.  The TREAT main effect is
represented by CTREAT with one degree of freedom.
CTRGR1 and CTRGR2 are simply cross products of
the other main effects and test the GRADE X TREAT
interaction effects.  Table 5 presents results for the
ANOVA: GRADE, F = 19.367; TREAT, F = .510;

Table 4 . Conducting t-test and Point-biserial
Correlation with Canonical (EFFORT by TREAT)

t-test Canonical Point-biserial
t(10) .314 F(1, 10) .100

p .760 p .760 p .760
M(TREAT1) 12.500
SD(TREAT1) 4.848 R c .100 r -.100
M(TREAT2) 11.667 Rc

2 .010 r2 .010
SD(TREAT2) 4.320 lambda .990
Note.  Rc cannot be negative.

GRADE X TREAT, F = 3.449.  Note that the effect
size (r2) for the error term was .1323.

Obtaining comparable results with CCA requires
us to take several steps.  The first step involves
conducting canonical analyses in four separate designs,
using EFFORT as the dependent measure and the
contrasts as independent variables.   Design 1 included
all planned contrasts, CGR1, CGR2, CTREAT,
CTRGR1, and CTRGR2, to test the total effect (SOS
explained).  Design 2 used CTREAT, CTRGR1, and
CTRGR2 to jointly test the TREAT and interaction
effects.  Design 3 used CGR1, CGR2, CTRGR1, and
CTRGR2 to jointly test the GRADE and interaction
effects.  The final CCA, Design 4, used CGR1,
CGR2, and CTREAT to jointly test the GRADE and
TREAT effects.  Table 6 displays the Wilks’ lambda
values for each design from the first step.  Remember
that λ  is something of a “reverse” effect size and will
equal the effect for the error term.  A quick comparison
of λ  for the total effect (Table 6) with the error effect
size (Table 5) confirms this relationship between the
statistics.

After canonical lambdas have been attained, we
must use them to determine the omnibus ANOVA
lambdas.  This was done by dividing the Design 1
total effect (lambda) by the lambdas of the other
designs.  For example, to find the omnibus lambda for
the GRADE main effect the total lambda (.11507) was
divided by the Design 2 lambda (.85793), which
reflects the joint effect of the contrast variables for the
TREAT main effect and the GRADE X TREAT
interaction effect.  This process “removes” the effects
of the other hypotheses, leaving the omnibus lambda
for the GRADE main effect to be .13412516 (.11507 /
.85793 = .13412516 = λ).  The same process was used
to find the other ANOVA lambdas with results
reported in Table 6.

One final step remained.  ANOVA lambdas were
converted into ANOVA F statistics using the
following formula: [(1 – λ)/λ]*(dferror / df effect) = F.

To illustrate, the F value for the GRADE main
effect was modeled by [(1 - .13413) / .13413] * (6 / 2)
= 19.36636.  Table 6 also reports transformations for
both main effects and the interaction.  Note that the F
statistics obtained by the canonical process match

Table 5.  3 X 2 Factorial ANOVA
(EFFORT by GRADE and TREAT)

Source SS df MS F p eta2

GRADE 158.167 2 79.083 19.367 .002 .743
TREAT 2.083 1 2.083 0.510 .502 .010
G x T 28.167 2 14.083 3.449 .101 .132
Error 24.500 6
Total 212.917 11

those obtained by the factorial ANOVA (see Table 5),
within rounding error of course.

It should also be noted that the equivalence of
ANOVA and CCA can be demonstrated with dummy
codes that represent group membership in the
independent variable (see Fan, 1978).  However, the
predictors would be correlated in this case.  The use of
orthogonal contrast is useful here to maintain the
factorial structure of the groups.

Conducting Factorial MANOVA
with Canonical Correlation

Since SPSS can use the MANOVA command to
perform CCA, it would seem that the two are related.
To illustrate the relationship, a 3 X 2 factorial
MANOVA was computed with EFFORT and ABILIT
as dependent variables and GRADE and TREAT as
independent measures.  Results from this analysis are
found in Table 7.  Since MANOVA is a multivariate
method, Wilks lambdas are reported by SPSS and are
used to test statistical significance of the F values.    

The comparable canonical analysis was
performed using the same process as with the ANOVA
above.  Four CCA designs using the contrast variables
were run with canonical lambdas reported in Table 8.
The subsequent conversion of these values to
MANOVA lambdas is also found in Table 8.  The
reader will note the equivalence of the MANOVA _s in
Table 7 with those obtained through the canonical
analysis in Table 8.  The final conversion to F values
was not necessary here since the MANOVA uses the
λ value to calculate F statistics, unlike the SOS value
used in ANOVA.  However, the same full model F-
test formula used in the ANOVA section can be used
to find the F statistics in this case.  

Conducting Discriminant Analysis
with Canonical Correlation

Discriminant analysis is a multivariate method
that can either be used predictively to classify persons
into groups or descriptively where variables identify
latent structures among groups (Huberty, 1994).  The
descriptive discriminant analysis (DDA) case is
especially useful as the preferred substitute for a one-
way MANOVA or as a post hoc analysis to multi-way
MANOVA analyses.

Table 6. Conduct ANOVA with Canonical Analysis
            (EFFORT by Contrasts)

Step One: Canonical Analyses on Four Designs
Design Independent Variables lambda

1 CGR1, CGR2, CTREAT
    CTGR1, CTGR2

.11507

2 CTREAT, CTGR1, CTGR2 .85793
3 CGR1, CGR2, CTGR1, CTGR2 .12485
4 CGR1, CGR2, CTREAT .24736

Step Two: Conversion of Canonical Lambdas to
                     ANOVA Lambdas

ANOVA
Effect Designs Transformation

ANOVA
Lambda

GRADE 1 / 2 .11507/.85793 .13412516
TREAT 1 / 3 .11507/.12485 .92166600
G x T 1 / 4 .11507/.24736 .46519243
Step Three: Conversion of ANOVA Lambdas to
                    F-ratio

Source Transformation F-ratio
GRADE [(1-.13413)/.13413]*(6/2) 19.36636
TREAT [(1-.92167)/.92167]*(6/1) 0.50992
G x T [(1-.46519)/.46519]*(6/2) 3.44898

To demonstrate the DDA and CCA relationship, a
descriptive discriminant analysis was conducted with
TREAT as the nominally scaled predictor variable and
EFFORT and ABILIT as criterion variables.  Table 9
reports a statistically non-significant result χ2(2, 9) =
.648, p = .723.  The canonical analysis was conducted
using the planned contrast variable CTREAT as the
predictor.  Results of the CCA are also reported in
Table 9.  The reader will note that the analyses yield
identical results.  One arbitrary difference is in the
reporting of a χ2 statistic for the discriminant analysis
as opposed to the CCA F value.  As with the t and F
distributions described above, the difference is arbitrary
since the χ2 and F statistics represent the same value
expressed in a different metric.  The χ2 statistic can be
calculated by multiplying the F value by (j * k), where
j is the number of variables in the predictor set and k
is the number of variables in the criterion set.  In this
case, F = .33602, so χ2 = (1 * 2).33602 = .67204.
This transformation approximates the χ2 reported in
Table 9 with the difference due to rounding.

Table 9. Conducting Multiple Regression with
Canonical (EFFORT by LOCUS and ABILIT)

Discriminant Analysis Canonical Analysis
R c .264 R c .264
Rc

2 .070 Rc
2 .070

lambda .931 lambda .931
.648 F .336

df 2, 9 df 2, 9
p .723 p .723

Table 7.   3 X 2 Factorial ANOVA
(EFFORT and ABILIT by GRADE and TREAT)
Source lambda df F p

GRADE .05061 4, 10 8.61299 .003
TREAT .61798 2,  5 1.54541 .300
G x T .44653 4, 10 1.24122 .354

Table 8.  Conduct MANOVA with Canonical
   Analysis (EFFORT and ABILIT by Contrasts)
Step One: Canonical Analyses on Four Designs
Design Independent Variables lambda

1 CGR1, CGR2, CTREAT
    CTGR1, CTGR2

.03184

2 CTREAT, CTGR1, CTGR2 .62924
3 CGR1, CGR2, CTGR1, CTGR2 .05153
4 CGR1, CGR2, CTREAT .07132

Step Two: Conversion of Canonical Lambdas to
                     MANOVA Lambdas

ANOVA
Effect Designs Transformation

ANOVA
Lambda

GRADE 1 / 2 .03184/.62924 .05060072
TREAT 1 / 3 .03184/.05153 .61789249
G x T 1 / 4 .03184/.07132 .44643859

Conclusion
The purpose of the present article has been to

illustrate that canonical correlation analysis represents
the multivariate parametric general linear model.  As
such, CCA can be used to conduct the univariate and
multivariate analyses that CCA subsumes, including
multiple regression. The point is heuristic and not in-
tended to suggest that all analyses should be conduct-
ed with CCA.  In fact, it is quite clear in the ANOVA
and MANOVA examples that CCA, at least as
reported by SPSS, is the long way to the same results.
However, CCA would be superior to ANOVA and
MANOVA when the independent variables are
intervally scaled, thus eliminating the need to discard
variance.  

Knowing that there is a general linear model and
understanding that all parametric analyses are
intricately related can be of great educational value to
both students and teachers of quantitative methods as
well as practicing researchers.  Knowing these
relationships facilitates understanding of commonalties
and differences among all the parametric methods and
serves to inform researcher judgement concerning
analysis selection and use.

The author would like to graciously thank Bruce
Thompson for a review of an earlier draft of the
manuscript.

Correspondence should be addressed to
Robin K. Henson
Department of Educational Leadership & Research
University of Southern Mississippi
Email: rrhenson@aol.com.



Canonical Analysis

Multiple Linear Regression Viewpoints, 2000, Vol. 26(1) 17

References
Baggaley, A. R. (1981). Multivariate analysis: An

introduction for consumers of behavioral research.
Evaluation Review, 5, 123-131.

Bagozzi, R. P., Fornell, C., & Larcker, D. F. (1981).
Canonical correlation analysis as a special case of a

structural relations model. Multivariate Behavioral
Research, 16, 437-454.

Campbell, K. T., & Taylor, D. L. (1996). Canonical
correlation analysis as a general linear model: A
heuristic lesson for teachers and students. Journal of
Experimental Education, 64, 157-171.

Cliff, N. (1987). Analyzing multivariate data. San
Diego,Harcourt Brace Jovanovich.

Cohen, J. (1968). Multiple regression as a general
data-analytic system. Psychological Bulletin, 70,
426-443.

Edgington, E. S. (1974). A new tabulation of
statistical procedures used in APA journals.
American Psychologist, 29, 25-26.

Elmore, P. B., & Woehlke, P. L. (1988).  Statistical
methods employed in American Educational
Research Journal, Educational Researcher, and
Review of Educational Research from 1978 to
1987. Educational Researcher, 17(9), 19-20.

Fan, X. (1996). Canonical correlation analysis as a
general analytic model. In B. Thompson (Ed.),
Advances in social science methodology (Vol. 4,
pp. 71-94). Greenwich, CT: JAI Press.

Fan, X. (1997). Canonical correlation analysis and
structural equation modeling: What do they have in
common? Structural Equation Modeling, 4, 65-79.

Fish, L. J. (1988). Why multivariate methods are
usually vital. Measurement and Evaluation in
Counseling and Development, 21, 130-137.

Fornell, C. (1978). Three approaches to canonical
analysis. Journal of the Market Research Society,
20, 166-181.

Goodwin, L. D., & Goodwin, W. L. (1985).
Statistical techniques in AERJ articles, 1979-1983:
The preparation of graduate students to read the
educational research literature. Educational
Researcher, 14(2), 5-11.

Henson, R. K. (in press). Multivariate normality:
What is it and how is it assessed? In B. Thompson
(Ed.), Advances in social science methodology (Vol.
5). Stamford, CT: JAI Press.

Hotelling, H. (1935). The most predictable criterion.
Journal of Experimental Psychology, 26, 139-142.

Huberty, C. (1994). Applied discriminant analysis.
New York: Wiley.

Humphreys, L. G. (1978). Doing research the hard
way: Substituting analysis of variance for a
problem in correlational analysis. Journal of
Educational Psychology, 70, 873-876.

Humphreys, L. G., & Fleishman, A. (1974). Pseudo-
orthogonal and other analysis of variance designs
involving individual-differences variables. Journal of
Educational Psychology, 66, 464-472.

Knapp, T. R. (1978). Canonical correlation analysis:
A general parametric significance-testing system.
Psychological Bulletin, 85, 410-416.

Krus, D. J., Reynolds, T. S., & Krus, P. H. (1976).
Rotation in canonical variate analysis. Educational
and Psychological Measurement, 36, 725-730.

Sexton, J. D., McLean, M., Boyd, R. D., Thompson,
B., & McCormick, K. (1988). Criterion-related
validity of a new standardized measure for use with
infants who are handicapped. Measurement and
Evaluation in Counseling and Development, 21,
16-24.

Thompson, B. (in press). Canonical correlation
analysis. In L. Grimm & P. Yarnold (Eds.),
Reading and understanding multivariate statistics
(Vol. 2). Washington, DC: American Psychological
Association.

Thompson, B. (1991). A primer on the logic and use
on canonical correlation analysis. Measurement and
Evaluation in Counseling and Development, 24(2),
80-95.

Thompson, B. (1994). Planned versus unplanned and
orthogonal versus nonorthongonal contrasts: The
neo-classical perspective. In B. Thompson (Ed.),
Advances in social science methodology (Vol. 3,
pp. 3-27). Greenwich, CT: JAI Press.

Thompson (1998, April). Five methodology errors in
educational research: The pantheon of statistical
significance and other faux pas. Invited address
presented at the annual meeting of the American
Educational Research Association, San Diego.
(Internet URL   http://acs.tamu.edu/~bbt6147/  )

Willson, V. L. (1980). Research techniques in AERJ
articles: 1969 to 1978. Educational Researcher,
9(6), 5-10.



Henson

Multiple Linear Regression Viewpoints, 2000, Vol. 26(1)18

APPENDIX

SPSS Command Syntax for Canonical Demonstration

TITLE ' Canonical correlation
        demonstration '.
TITLE ' Robin K. Henson '.
COMMENT Heuristic data for 12 cases
COMMENT EFFORT - attributions of
              effort
COMMENT ABILIT - attributions of
             ability
COMMENT LOCUS - external vs internal
            locus of control
COMMENT EXTROV - degree of
          extroversion scale
COMMENT GRADE - elementary(1),
         middle(2), high(3) school
COMMENT TREAT - treat(1),
         control(2) groups.
SET BLANKS=SYSMIS UNDEFINED=WARN
PRINTBACK LISTING.
DATA LIST
  FILE='c:\ccaasglm.txt'
  FIXED RECORDS=1
  /ID 1-2 EFFORT 4-5 ABILIT 7-8
  LOCUS 10-11 EXTROV 13-14
  GRADE 16 TREAT 18.
EXECUTE.
COMMENT Show that cca can do Pearson r.
CORRELATIONS
 /VARIABLES=EFFORT ABILIT
 /PRINT=TWOTAIL NOSIG
 /MISSING=PAIRWISE .
MANOVA
 EFFORT WITH ABILIT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).
COMMENT Show that cca can do
      multiple regression.
REGRESSION
 /MISSING LISTWISE
 /STATISTICS COEFF OUTS R ANOVA
 /CRITERIA=PIN(.05) POUT(.10)
 /NOORIGIN
 /DEPENDENT EFFORT
 /METHOD=ENTER LOCUS EXTROV  .
MANOVA
 LOCUS EXTROV WITH EFFORT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).
COMMENT Show that cca can do t-test
   and point biserial correlation.
T-TEST
 GROUPS=TREAT(1 2)
 /MISSING=ANALYSIS
 /VARIABLES=EFFORT
 /CRITERIA=CIN(.95) .

MANOVA
 TREAT WITH EFFORT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).
COMMENT Show cca can do point-biserial
    which is a generalization of r.
CORRELATIONS
 /VARIABLES = treat effort
 /PRINT=TWOTAIL NOSIG
 /MISSING=PAIRWISE .
COMMENT Show that cca can do
      factorial ANOVA.
COMMENT Compute contrast variables
      to do cca.
IF (GRADE = 1) CGR1 = -1.
IF (GRADE = 2) CGR1 = 0.
IF (GRADE = 3) CGR1 = 1.
COMMENT Tests equality of the
   means of elementary(4) vs
    high school(4) students.
EXECUTE.
IF (CGR1 = -1) CGR2 = -1.
IF (CGR1 = 0) CGR2 = 2.
IF (CGR1 = 1) CGR2 = -1.
EXECUTE.
COMMENT Tests equality of means
    of middle(4) vs
    elementary high school(8) students.
IF (TREAT = 1) CTREAT = -1.
IF (TREAT = 2) CTREAT = 1.
EXECUTE.
COMMENT Tests equality of means of
   treatment (6) vs control groups (6).
COMPUTE CTRGR1 = CGR1 * CTREAT.
COMPUTE CTRGR2 = CGR2 * CTREAT.
EXECUTE.
COMMENT Tests treatment by grade
    interaction effects.
COMMENT Show contrast variables
     are orthogonal.
CORRELATIONS
 /VARIABLES=CGR1 CGR2 CTREAT
           CTRGR1 CTRGR2
  /PRINT=TWOTAIL SIG
  /MISSING=PAIRWISE .
COMMENT Step one: run factorial ANOVA
       and cca on constrast variables.
ANOVA
  VARIABLES=EFFORT
  BY GRADE(1 3) TREAT(1 2)
  /MAXORDERS ALL
  /METHOD UNIQUE
  /FORMAT LABELS .

MANOVA
 CGR1 CGR2 CTREAT CTRGR1 CTRGR2
   WITH EFFORT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).
MANOVA
 CTREAT CTRGR1 CTRGR2 WITH EFFORT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).
MANOVA
 CGR1 CGR2 CTRGR1 CTRGR2 WITH EFFORT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).
MANOVA
 CGR1 CGR2 CTREAT WITH EFFORT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).
COMMENT Show cca can do MANOVA.
MANOVA
  EFFORT ABILIT BY GRADE(1 3)
        TREAT(1 2)
  /PRINT SIGNIF(MULT UNIV )
  /NOPRINT PARAM(ESTIM)
  /METHOD=UNIQUE
  /ERROR WITHIN+RESIDUAL
  /DESIGN  .
MANOVA
 CGR1 CGR2 CTREAT CTRGR1 CTRGR2
   WITH EFFORT ABILIT
 /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
 /DISCRIM=(STAN ESTIM COR).

MANOVA
  CTREAT CTRGR1 CTRGR2
    WITH EFFORT ABILIT
  /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
  /DISCRIM=(STAN ESTIM COR).
MANOVA
  CGR1 CGR2 CTRGR1 CTRGR2 WITH EFFORT
ABILIT
  /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
  /DISCRIM=(STAN ESTIM COR).
MANOVA
  CGR1 CGR2 CTREAT WITH EFFORT ABILIT
  /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
  /DISCRIM=(STAN ESTIM COR).
COMMENT Show cca can do
       discriminant analysis.
DISCRIMINANT
  /GROUPS=TREAT(1 2)
  /VARIABLES=EFFORT ABILIT
  /ANALYSIS ALL
  /PRIORS  EQUAL
  /CLASSIFY=NONMISSING POOLED .
MANOVA
  EFFORT ABILIT WITH CTREAT
  /PRINT=SIGNIF (MULTIV EIGEN DIMENR)
  /DISCRIM=(STAN ESTIM COR).

